



INSTITUTO FEDERAL DE  
EDUCAÇÃO, CIÊNCIA E TECNOLOGIA  
RIO GRANDE DO NORTE

# COMANDO DE SELEÇÃO MÚLTIPLA

Disciplina: Algoritmos

Professora: Alba Lopes

[alba.lopes@ifrn.edu.br](mailto:alba.lopes@ifrn.edu.br)

<http://docente.ifrn.edu.br/albalopes>

# COMANDO DE SELEÇÃO MÚLTIPLA

- Em alguns problemas, há múltiplas opções de decisão que devem ser tomadas
- Exemplos:
  - Crie um algoritmo que dado um número de 1 a 7, informe o dia da semana correspondente
    - É possível construir o algoritmo utilizando os comandos se...então



# EXEMPLO 1

```
algoritmo "dia_da_semana"  
var  
    dia: inteiro  
inicio  
    escreva("Digite um número de 1 a 7:")  
    leia(dia)  
    se dia = 1 entao  
        escreva("Domingo")  
    senao  
        se dia = 2 entao  
            escreva ("Segunda-feira")  
        senao  
            se dia = 3 entao  
                escreva("Terça-feira")  
            senao  
                se dia = 4 entao  
                    escreva("Quarta-feira")  
                senao  
                    se dia = 5 entao  
                        escreva("Quinta-feira")  
                    senao  
                        se dia = 6 entao  
                            escreva("Sexta-feira")  
                        senao  
                            se dia = 7 entao  
                                escreva("Sábado")  
                            senao  
                                escreva("Dia da semana inválido!")  
                            fimse  
                        fimse  
                    fimse  
                fimse  
            fimse  
        fimse  
    fimse  
fimalgoritmo
```



# COMANDO DE SELEÇÃO MÚLTIPLA

- O comando de Seleção Múltipla facilita a solução de problemas desse tipo

```
escolha <expressao-de-selecao>  
caso <valor11>, <valor12>, ..., <valor1N>  
    <sequencia-de-comandos-1>  
caso <valor21>, <valor22>, ..., <valor2N>  
    <sequencia-de-comandos-2>  
...
```



## EXEMPLO 1

```
algoritmo "dia_da_semana"  
var  
    dia: inteiro  
inicio  
    escreva("Digite um número de 1 a 7:")  
    leia(dia)  
    escolha dia  
    caso 1  
        escreva("Domingo")  
    caso 2  
        escreva ("Segunda-feira")  
    caso 3  
        escreva("Terça-feira")  
    caso 4  
        escreva("Quarta-feira")  
    caso 5  
        escreva("Quinta-feira")  
    caso 6  
        escreva("Sexta-feira")  
    caso 7  
        escreva("Sábado")  
    outrocaso  
        escreva("Dia inválido")  
    fimescolha  
fimalgoritmo
```



## EXEMPLO 2

### ○ Exemplo:

- Um restaurante oferece algumas opções de pratos do cardápio. Cada prato possui um valor, de acordo com a tabela:

Prato	Preço
Feijoada	R\$ 25,00
Macarronada	R\$ 15,00
Galinha caipira	R\$ 30,00
Carne de sol	R\$ 30,00

- Crie um algoritmo que, dado o nome do prato, exiba a mensagem de quanto o cliente irá pagar



## EXEMPLO 2

```
algoritmo "cardapio"  
var  
    prato: caractere  
inicio  
    escreval("-Cardapio-")  
    escreval("Feijoadada")  
    escreval("Macarronada")  
    escreval("Galinha caipira")  
    escreval("Carne de sol")  
    escreva("Digite o nome do prato que você deseja pedir:")  
    leia(prato)  
    escolha prato  
    caso "Feijoadada"  
        escreval("O valor da conta é: R$ 25,00")  
    caso "Macarronada"  
        escreval("O valor da conta é: R$ 15,00")  
    caso "Galinha caipira", "Carne de sol"  
        escreval("O valor da conta é: R$ 30,00")  
    outrocaso  
        escreva("Prato indisponível no cardápio. Volte sempre!")  
    fimescolha  
fimalgoritmo
```



# EXERCÍCIO 1

- Uma loja está fazendo liquidação e vendendo produtos com até 50% de desconto. O desconto é identificado de acordo com a cor da etiqueta do produto:

Cor da etiqueta	Desconto
azul	10%
verde	20%
amarelo	30%
vermelho	40%
laranja	50%

- Crie um algoritmo que leia o valor original do produto e a cor da etiqueta e informe quanto o cliente irá pagar após a aplicação do desconto

# LISTA DE EXERCÍCIOS II

- Atividade Individual
- Entregar por **e-mail** até **28/05/2012**
- **Valor: 1,0 ponto**
- Disponível no site da disciplina:
  - <http://docente.ifrn.edu.br/albalopes/>

